

# Chapter 45

## INSET and INSET2 Statements

### Chapter Table of Contents

---

<b>OVERVIEW</b> . . . . .	1595
<b>GETTING STARTED</b> . . . . .	1596
Displaying Summary Statistics on a Control Chart . . . . .	1596
Formatting Values and Customizing Labels . . . . .	1598
Adding a Header and Positioning the Inset . . . . .	1599
<b>SYNTAX</b> . . . . .	1601
Summary of INSET Keywords . . . . .	1602
Summary of Options . . . . .	1604
Dictionary of Options . . . . .	1605
<b>DETAILS</b> . . . . .	1607
Positioning the Inset Using Compass Points . . . . .	1607
Positioning the Inset in the Margins . . . . .	1608
Positioning the Inset Using Coordinates . . . . .	1608



## Chapter 45

# INSET and INSET2 Statements

---

### Overview

The INSET and INSET2 statements enable you to enhance a Shewhart chart by adding a box or table (referred to as an *inset*) of summary statistics directly to the graph. The INSET statement places an inset in a primary Shewhart chart while the INSET2 statement places one in a secondary Shewhart chart. An inset can display statistics calculated by the SHEWHART procedure or arbitrary values provided in a SAS data set.

Note that an INSET or INSET2 statement by itself does not produce a display but must be used in conjunction with a chart statement. Insets are not available with line printer output, so the INSET and INSET2 statements are not applicable when the LINEPRINTER option is specified in the PROC SHEWHART statement.

You can use options in the INSET and INSET2 statements to

- specify the position of the inset
- specify a header for the inset table
- specify graphical enhancements, such as background colors, text colors, text height, text font, and drop shadows

The INSET2 statement differs from the INSET statement in only two respects.

1. An INSET2 statement creates an inset within a secondary chart generated by an IRCHART, MRCHART, XRCHART or XSCHART statement or by the TRENDVAR= option. For example, when following an XRCHART statement an INSET statement produces an inset in the  $\bar{X}$  chart and an INSET2 statement produces one in the  $R$  chart.
2. The INSET statement can be used to place an inset in one of the margins surrounding the plot area, while the INSET2 statement cannot.

Any of the statistics available for display in an inset can be specified with either an INSET or INSET2 statement. Descriptions of the INSET statement in this chapter also apply to the INSET2 statement except where explicitly noted.

---

## Getting Started

This section introduces the INSET statement with examples that illustrate commonly used options. Complete syntax for the INSET statement is presented in the “Syntax” section on page 1601.

---

## Displaying Summary Statistics on a Control Chart

In the manufacture of silicon wafers, batches of five wafers are sampled, and their diameters are measured in millimeters. The following statements create a SAS data set named WAFERS, which contains the measurements for 25 batches:

```

data wafers;
  input batch @;
  do i=1 to 5;
    input diamtr @;
    output;
  end;
  drop i;
datalines;
1 35.00 34.99 34.99 34.98 35.00
2 35.01 34.99 34.99 34.98 35.00
3 34.99 35.00 35.00 35.00 35.00
4 35.01 35.00 34.99 34.99 35.00
5 35.00 34.99 34.98 34.99 35.00
6 34.99 34.99 35.00 35.00 35.00
7 35.01 34.98 35.00 35.00 34.99
8 35.00 35.00 34.99 34.98 34.99
9 34.99 34.98 34.98 35.01 35.00
10 34.99 35.00 35.01 34.99 35.01
11 35.01 35.00 35.00 34.98 34.99
12 34.99 34.99 35.00 34.98 35.01
13 35.01 34.99 34.98 34.99 34.99
14 35.00 35.00 34.99 35.01 34.99
15 34.98 34.99 34.99 34.98 35.00
16 34.99 35.00 35.00 35.01 35.00
17 34.98 34.98 34.99 34.99 34.98
18 35.01 35.02 35.00 34.98 35.00
19 34.99 34.98 35.00 34.99 34.98
20 34.99 35.00 35.00 34.99 34.99
21 35.00 34.99 34.99 34.98 35.00
22 35.00 35.00 35.01 35.00 35.00
23 35.02 35.00 34.98 35.02 35.00
24 35.00 35.00 34.99 35.01 34.98
25 34.99 34.99 34.99 35.00 35.00
;

```

The following statements generate an  $\bar{X}$  chart from the WAFERS data. Lower and upper specification limits for wafer diameters are given and the process capability index  $C_p$  is computed. An INSET statement is used to display the specification limits, the computed value of  $C_p$  and the process standard deviation on the chart:


```

title 'Mean Chart with Inset';
proc shewhart data=wafers;
  xchart diamtr*batch /
    lsl = 34.97
    usl = 35.03;
  inset lsl usl cp stddev /
    height = 3;
run;

```

The resulting  $\bar{X}$  chart is displayed in Figure 45.1. The INSET statement immediately follows the chart statement that creates the graphical display (in this case, the XCHART statement). Specify the keywords for inset statistics (such as LSL, USL, CP and STDDEV) immediately after the word INSET. The inset statistics appear in the order in which you specify the keywords. The HEIGHT= option on the INSET statement specifies the text height used to display the statistics in the inset.

A complete list of keywords that you can use with the INSET statement is provided in “Summary of INSET Keywords” on page 1602. Note that the set of keywords available for a particular display depends on both the plot statement that precedes the INSET statement and the options that you specify in the plot statement.



**Figure 45.1.** An  $\bar{X}$  Chart with an Inset

The following examples illustrate options commonly used for enhancing the appearance of an inset.

---

## Formatting Values and Customizing Labels

By default, each inset statistic is identified with an appropriate label, and each numeric value is printed using an appropriate format. However, you may want to provide your own labels and formats. For example, in Figure 45.1 the default format used for  $C_p$  and the process standard deviation prints an excessive number of decimal places. The following statements produce  $\bar{X}$  and  $R$  charts, each with its own inset. The unwanted decimal places are eliminated and the default specification limits labels are replaced with abbreviations:

```
proc shewhart data=wafers;
  xrchart diamtr*batch /
    lsl = 34.97
    usl = 35.03;
  inset lsl='LSL' usl='USL' /
    height = 3;
  inset2 cp (6.4) stddev (6.4) /
    height = 3;
run;
```

The resulting  $\bar{X}$  and  $R$  charts are displayed in Figure 45.2. You can provide your own label by specifying the keyword for that statistic followed by an equal sign (=) and the label in quotes. The label can have up to 24 characters.

The format 6.4 specified in parentheses after the CP and STDDEV keywords displays those statistics with a field width of six and four decimal places. In general, you can specify any numeric SAS format in parentheses after an inset keyword. You can also specify a format to be used for all the statistics in the INSET statement with the FORMAT= option. For more information about SAS formats, refer to Chapter 14 of *SAS Language Reference: Dictionary*.

Note that if you specify both a label and a format for a statistic, the label must appear before the format.




Figure 45.2. Formatting Values and Customizing Labels in an Inset

## Adding a Header and Positioning the Inset

In the previous examples, the insets are displayed in the upper left corners of the plots, the default position for insets added to control charts. You can control the inset position with the `POSITION=` option. In addition, you can display a header at the top of the inset with the `HEADER=` option. The following statements create a data set to be used with the `INSET DATA=` keyword and the chart shown in Figure 45.3:

```
data location;
  length _LABEL_ $ 10 _VALUE_ $ 12;
  input _LABEL_ _VALUE_ &;
datalines;
Plant      Santa Clara
Line       1
Shift      2
;

proc shewhart data=wafers;
  xchart diamtr*batch /
    lsl = 34.97
    usl = 35.03;
  inset data= location lsl='LSL' usl='USL' cp (6.4) stddev (6.4) /
    position = rm
    cshadow  = black
    header   = 'Summary Statistics';
run;
```

Part 9. The CAPABILITY Procedure

The header (in this case, *Summary Statistics*) can be up to 40 characters. Note that a relatively long list of inset statistics is requested. Consequently, POSITION=RM is specified to position the inset in the right margin. For more information about positioning, see “Details” on page 1607. The CSHADOW= option is used to display a drop shadow on this inset. The options, such as HEADER=, POSITION= and CSHADOW= are specified after the slash (/) in the INSET statement. For more details on INSET statement options, see “Dictionary of Options” on page 1605.

Note that the contents of the data set LOCATION appear before other statistics in the inset. The position of the DATA= keyword in the keyword list determines the position of the data set’s contents in the inset.




Figure 45.3. Adding a Header and Repositioning the Inset



## Syntax

The syntax for the INSET and INSET2 statements is as follows:

```
INSET keyword-list < / options >;
INSET2 keyword-list < / options >;
```

You can use any number of INSET and INSET2 statements in the SHEWHART procedure. Each INSET or INSET2 statement produces a separate inset and must follow one of the chart statements. The inset appears on every panel (page) produced by the last chart statement preceding it. The statistics are displayed in the order in which they are specified. The following statements produce a boxplot with two insets and an  $\bar{X}$  and  $R$  chart with one inset in the  $\bar{X}$  chart and one in the  $R$  chart.

```
proc shewhart data=wafers;
  boxchart diamtr * batch / lsl=34.9 target=35 usl=35.1;
    inset lsl target usl;
    inset cp cpk cpm;
  xrchart diamtr*batch;
    inset nmin nmax nout;
    inset2 nlow2 nhigh2;
run;
```

The statistics displayed in an inset are computed for a specific process variable using observations for the current BY group. For example, in the following statements, there are two process variables (WEIGHT and DIAMETER) and a BY variable (LOCATION). If there are three different locations (levels of LOCATION), then a total of six  $\bar{X}$  charts are produced. The statistics in each inset are computed for a particular variable and location. The labels in the inset are the same for each  $\bar{X}$  chart.

```
proc shewhart data=axles;
  by location;
  xchart (weight diameter) * batch / tests=1 to 8;
  inset ntests 1 to 8;
run;
```

The components of the INSET and INSET2 statements are described as follows.

### *keyword-list*

can include any of the *keywords* listed in “Summary of INSET Keywords” on page 1602. Some *keywords*, such as NTESTS and DATA=, require operands specified immediately after the *keyword*. Also, some inset statistics are available only if you request chart statements and options for which those statistics are calculated. For example,

- the NHIGH2, NLOW2, NTESTS2, LCL2 and UCL2 keywords are available only when a secondary chart is produced with the IRCHART, MRCHART, XRCHART or XSCHART statements.
- the NTESTS *keyword* requires the TESTS= option;
- the NTESTS2 *keyword* requires the TESTS2= option;

- the capability index *keywords* such as CPK all require one or more of the LSL=, USL= and TARGET= options.

By default, inset statistics are identified with appropriate labels, and numeric values are printed using appropriate formats. However, you can provide customized labels and formats. You provide the customized label by specifying the *keyword* for that statistic followed by an equal sign (=) and the label in quotes. Labels can have up to 24 characters. You provide the numeric format in parentheses after the *keyword*. Note that if you specify both a label and a format for a statistic, the label must appear before the format. For an example, see “Formatting Values and Customizing Labels” on page 1598.

#### *options*

appear after the slash (/) and control the appearance of the inset. For example, the following INSET statement uses two appearance *options* (POSITION= and CTEXT=):

```
inset n nmin nmax / position=ne ctext=yellow;
```

The POSITION= option determines the location of the inset, and the CTEXT= option specifies the color of the text of the inset.

See “Summary of Options” on page 1604 for a list of all available *options*, and “Dictionary of Options” on page 1605 for detailed descriptions. Note the difference between *keywords* and *options*; *keywords* specify the information to be displayed in an inset, whereas *options* control the appearance of the inset.

---

## Summary of INSET Keywords

All keywords available with the SHEWHART procedure’s INSET and INSET2 statements request a single statistic in an inset, except for the NTESTS, NTESTS2 and DATA= keywords. The NTESTS and NTESTS2 keywords each require a list of indexes specifying the tests for special causes whose counts of positive results are to be displayed:

```
inset ntests 1 2 3 4;  
inset ntests2 1 to 4;
```

For each of the requested tests, the number of positive results for the test is displayed in the inset. So if tests 1 through 4 are requested the results occupy four lines in the inset.

The DATA= keyword specifies a SAS data set containing (label, value) pairs to be displayed in an inset. The data set must contain the variables \_LABEL\_ and \_VALUE\_. \_LABEL\_ is a character variable whose values provide labels for inset entries. \_VALUE\_ can be character or numeric, and provides values displayed in the inset. The label and value from each observation in the DATA= data set occupy one line in the inset. Figure 45.3 shows an inset containing entries from a DATA= data set.

**Table 45.1.** Summary Statistics

MEAN	estimated or specified process mean
N	nominal subgroup size
NMIN	minimum subgroup size
NMAX	maximum subgroup size
NOUT	number of subgroups outside control limits on primary chart
NLOW	number of subgroups below lower control limit on primary chart
NHIGH	number of subgroups above upper control limit on primary chart
NTESTS	number of positive results of tests for special causes on primary chart
STDDEV	estimated or specified process standard deviation
DATA=	(label, value) pairs from <i>SAS-data-set</i>

**Table 45.2.** Secondary Chart Summary Statistics

MEAN2	mean of subgroup ranges or standard deviations
NOUT2	number of subgroups outside control limits on secondary chart
NLOW2	number of subgroups below lower control limit on secondary chart
NHIGH2	number of subgroups above upper control limit on secondary chart
NTESTS2	number of positive results of tests for special causes on secondary chart

**Table 45.3.** Specification Limits

LSL	lower specification limit
USL	upper specification limit
TARGET	target value

**Table 45.4.** Capability Indices and Confidence Limits

CIALPHA	$\alpha$ value for computing capability index confidence limits
CP	capability index $C_p$
CPLCL	lower confidence limit for $C_p$
CPUCL	upper confidence limit for $C_p$
CPK	capability index $C_{pk}$
CPKLCL	lower confidence limit for $C_{pk}$
CPKUCL	upper confidence limit for $C_{pk}$
CPL	capability index $CPL$
CPLLCL	lower confidence limit for $CPL$
CPLUCL	upper confidence limit for $CPL$
CPM	capability index $C_{pm}$
CPMLCL	lower confidence limit for $C_{pm}$
CPMUCL	upper confidence interval for $C_{pm}$
CPU	capability index $CPU$
CPULCL	lower confidence limit for $CPU$
CPUUCL	upper confidence limit for $CPU$

---

## Summary of Options


The following table lists the INSET and INSET2 statement options. For complete descriptions, see “Dictionary of Options,” which follows this section.

**Table 45.5.** INSET Options

CFILL= <i>color</i>   BLANK	specifies color of inset background
CFILLH= <i>color</i>	specifies color of header background
CFRAME= <i>color</i>	specifies color of frame
CHEADER= <i>color</i>	specifies color of header text
CSHADOW= <i>color</i>	specifies color of drop shadow
CTEXT= <i>color</i>	specifies color of inset text
DATA	specifies data units for POSITION=( <i>x</i> , <i>y</i> ) coordinates
FONT= <i>font</i>	specifies font of text
FORMAT= <i>format</i>	specifies format of values in inset
HEADER= <i>'quoted string'</i>	specifies header text
HEIGHT= <i>value</i>	specifies height of inset text
NOFRAME	suppresses frame around inset
POSITION= <i>position</i>	specifies position of inset
REFPOINT=BR BL TR TL	specifies reference point of inset positioned with POSITION=( <i>x</i> , <i>y</i> ) coordinates

## Dictionary of Options

The following entries provide detailed descriptions of options for the INSET and INSET2 statements. Terms used in this section are illustrated in Figure 45.4.



**Figure 45.4.** The Inset

### **CFILL=***color* | **BLANK**

specifies the color of the background (including the header background if you do not specify the CFILLH= option).

If you do not specify the CFILL= option, then by default, the background is empty. This means that items that overlap the inset (such as subgroup data points or control limits) show through the inset. If you specify any value for the CFILL= option, then overlapping items no longer show through the inset. Specify CFILL=BLANK to leave the background uncolored and also to prevent items from showing through the inset.

### **CFILLH=***color*

specifies the color of the header background. By default, if you do not specify a CFILLH= color, the CFILL= color is used.

### **CFRAME=***color*

specifies the color of the frame. By default, the frame is the same color as the axis of the plot.

### **CHEADER=***color*

specifies the color of the header text. By default, if you do not specify a CHEADER= color, the CTEXT= color is used.

### **CSHADOW=***color*

### **CS=***color*

specifies the color of the drop shadow. See Figure 45.3 on page 1600 for an example. By default, if you do not specify the CSHADOW= option, a drop shadow is not displayed.

**CTEXT=***color*

**CT=***color*

specifies the color of the text. By default, the inset text color is the same as the other text on the plot.

**DATA**

specifies that data coordinates are to be used in positioning the inset with the POSITION= option. The DATA option is available only when you specify POSITION= (*x*, *y*), and it must be placed immediately after the coordinates (*x*, *y*). For details, see the entry for the POSITION= option or “Positioning the Inset Using Coordinates” on page 1608. See Figure 45.7 on page 1609 for an example.

**FONT=***font*

specifies the font of the text. By default, the font is SIMPLEX if the inset is located in the interior of the plot, and the font is the same as the other text displayed on the plot if the inset is located in the exterior of the plot.

**FORMAT=***format*

specifies a format for all the values displayed in an inset. If you specify a format for a particular statistic, then this format overrides the format you specified with the FORMAT= option.

**HEADER=** *'string'*

specifies the header text. The *string* cannot exceed 40 characters. If you do not specify the HEADER= option, no header line appears in the inset.

**HEIGHT=***value*

specifies the height of the text.

**NOFRAME**

suppresses the frame drawn around the text.

**POSITION=***position*

**POS=***position*

determines the position of the inset. The *position* can be a compass point keyword, a margin keyword, or a pair of coordinates (*x*, *y*). You can specify coordinates in axis percent units or axis data units. For more information, see “Details” on page 1607. By default, POSITION=NW, which positions the inset in the upper left (northwest) corner of the display.

**REFPOINT=**BR | BL | TR | TL

**RP=**BR | BL | TR | TL

specifies the reference point for an inset that is positioned by a pair of coordinates with the POSITION= option. Use the REFPOINT= option with POSITION= coordinates. The REFPOINT= option specifies which corner of the inset frame you want positioned at coordinates (*x*, *y*). The keywords BL, BR, TL, and TR represent bottom left, bottom right, top left, and top right, respectively. See Figure 45.8 on page 1610 for an example. The default is REFPOINT=BL.

If you specify the position of the inset as a compass point or margin keyword, the REFPOINT= option is ignored. For more information, see “Positioning the Inset Using Coordinates” on page 1608.

## Details

This section provides details on three different methods of positioning the inset using the POSITION= option. With the POSITION= option, you can specify

- compass points
- keywords for margin positions
- coordinates in data units or percent axis units

### Positioning the Inset Using Compass Points

You can specify the eight compass points N, NE, E, SE, S, SW, W, and NW as keywords for the POSITION= option. The following statements create the display in Figure 45.5, which demonstrates all eight compass positions. The default is NW.

```
proc shewhart data=wafers;
  xchart diamtr*batch / tests= 1 to 8;
  inset ntests 1 / height=3 cfill=blank header='NW' pos=nw;
  inset ntests 2 / height=3 cfill=blank header='N ' pos=n ;
  inset ntests 3 / height=3 cfill=blank header='NE' pos=ne;
  inset ntests 4 / height=3 cfill=blank header='E ' pos=e ;
  inset ntests 5 / height=3 cfill=blank header='SE' pos=se;
  inset ntests 6 / height=3 cfill=blank header='S ' pos=s ;
  inset ntests 7 / height=3 cfill=blank header='SW' pos=sw;
  inset ntests 8 / height=3 cfill=blank header='W ' pos=w ;
run;
```





Figure 45.5. Insets Positioned Using Compass Points

---

## Positioning the Inset in the Margins

Using the INSET statement you can also position an inset in one of the four margins surrounding the plot area using the margin keywords LM, RM, TM, or BM, as illustrated in Figure 45.6. The INSET2 statement cannot be used to produce an inset in a margin.



**Figure 45.6.** Positioning Insets in the Margins

For an example of an inset placed in the right margin, see Figure 45.3 on page 1600. Margin positions are recommended if a large number of statistics are listed in the INSET statement. If you attempt to display a lengthy inset in the interior of the plot, it is likely that the inset will collide with the data display.

---

## Positioning the Inset Using Coordinates

You can also specify the position of the inset with coordinates: POSITION= ( $x, y$ ). The coordinates can be given in axis percent units (the default) or in axis data units.

### Data Unit Coordinates

If you specify the DATA option immediately following the coordinates, the inset is positioned using axis data units. For example, the following statements place the bottom left corner of the inset at 6 on the horizontal axis and 34.985 on the vertical axis:

```
proc shewhart data=wafers;
  xchart diamtr*batch;
  inset n / header = 'Position=(6,34.985)'
           position = (6,34.985) data;
run;
```

The control chart is displayed in Figure 45.7. By default, the specified coordinates determine the position of the bottom left corner of the inset. You can change this reference point with the REFPOINT= option, as in the next example.






Figure 45.7. Inset Positioned Using Data Unit Coordinates

#### Axis Percent Unit Coordinates

If you do not use the DATA option, the inset is positioned using axis percent units. The coordinates of the bottom left corner of the display are (0,0), while the upper right corner is (100,100). For example, the following statements create a  $\bar{X}$  chart with two insets, both positioned using coordinates in axis percent units:

```
proc shewhart data=wafers;
  xchart diamtr*batch;
  inset nmin / position = (5,25)
           header   = 'Position=(5,25)'
           height   = 3
           cfill    = blank
           refpoint = tl;
  inset nmax / position = (95,95)
           header   = 'Position=(95,95)'
           height   = 3
           cfill    = blank
           refpoint = tr;
run;
```

The display is shown in Figure 45.8. Notice that the REFPOINT= option is used to determine which corner of the inset is to be placed at the coordinates specified with the POSITION= option. The first inset has REFPOINT=TL, so the top left corner of the inset is positioned 5% of the way across the horizontal axis and 25% of the way up the vertical axis. The second inset has REFPOINT=TR, so the top right corner of the inset is positioned 95% of the way across the horizontal axis and 95% of the way up the vertical axis. Note also that coordinates in axis percent units must be *between* 0 and 100.




Figure 45.8. Inset Positioned Using Axis Percent Unit Coordinates

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/QC<sup>®</sup> User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999. 1994 pp.

**SAS/QC<sup>®</sup> User's Guide, Version 8**

Copyright © 1999 SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-493-4

All rights reserved. Printed in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, by any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

**U.S. Government Restricted Rights Notice.** Use, duplication, or disclosure of the software by the government is subject to restrictions as set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

SAS<sup>®</sup> and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute in the USA and other countries. <sup>®</sup> indicates USA registration.

IBM<sup>®</sup>, ACF/VTAM<sup>®</sup>, AIX<sup>®</sup>, APPN<sup>®</sup>, MVS/ESA<sup>®</sup>, OS/2<sup>®</sup>, OS/390<sup>®</sup>, VM/ESA<sup>®</sup>, and VTAM<sup>®</sup> are registered trademarks or trademarks of International Business Machines Corporation. <sup>®</sup> indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.